

Functional Programming

Shin-Cheng Mu

FLOLAC 2022

0 To Begin With...

Prerequisites

If you have done the homework requested before this summer school, you should have familiarised yourself with

- values and types, and basic list processing,
- basics of type classes,
- defining functions by pattern matching,
- guards, case, local definitions by `where` and `let`,
- recursive definition of functions,
- and higher order functions.

Recommended Textbooks

- *Introduction to Functional Programming using Haskell* [Bir98]. My recommended book. Covers equational reasoning very well.
- *Programming in Haskell* [Hut16]. A thin but complete textbook.
- *Learn You a Haskell for Great Good!* [Lip11], a nice tutorial with cute drawings!
- *Real World Haskell* [OSG98].
- *Algorithm Design with Haskell* [BG20].

1 Definition and Proof by Induction

Total Functional Programming

- The next few lectures concerns inductive definitions and proofs of datatypes and programs.
- While Haskell provides allows one to define non-terminating functions, infinite data structures, for now we will only consider its total, finite fragment.
- That is, we temporarily

- consider only finite data structures,
- demand that functions terminate for all value in its input type, and
- provide guidelines to construct such functions.

- Infinite datatypes and non-termination can be modelled with more advanced theory, which we cannot cover in this course.

1.1 Induction on Natural Numbers

Recalling “Mathematical Induction”

- Let P be a predicate on natural numbers.
 - What is a predicate? Such a predicate can be seen as a function of type $\text{Nat} \rightarrow \text{Bool}$.
 - So far, we see Haskell functions as simple mathematical functions too.
 - However, Haskell functions will turn out to be more complex than mere mathematical functions later. To avoid confusion, we do not use the notation $\text{Nat} \rightarrow \text{Bool}$ for predicates.
- We’ve all learnt this principle of proof by induction: to prove that P holds for all natural numbers, it is sufficient to show that
 - $P\ 0$ holds;
 - $P\ (1 + n)$ holds provided that $P\ n$ does.

1.1.1 Proof by Induction

Proof by Induction on Natural Numbers

- We can see the above inductive principle as a result of seeing natural numbers as defined by the datatype ¹

```
data Nat = 0 | 1+ Nat .
```

- That is, any natural number is either 0, or $1 + n$ where n is a natural number.

¹Not a real Haskell definition.

- In this lecture, $\mathbf{1}_+$ is written in bold font to emphasise that it is a data constructor (as opposed to the function $(+)$, to be defined later, applied to a number 1).

A Proof Generator

Given $P\ 0$ and $P\ n \Rightarrow P\ (\mathbf{1}_+ n)$, how does one prove, for example, $P\ 3$?

$$\begin{aligned} & P\ (\mathbf{1}_+ (\mathbf{1}_+ (\mathbf{1}_+ 0))) \\ \Leftarrow & \{ P\ (\mathbf{1}_+ n) \Leftarrow P\ n \} \\ & P\ (\mathbf{1}_+ (\mathbf{1}_+ 0)) \\ \Leftarrow & \{ P\ (\mathbf{1}_+ n) \Leftarrow P\ n \} \\ & P\ (\mathbf{1}_+ 0) \\ \Leftarrow & \{ P\ (\mathbf{1}_+ n) \Leftarrow P\ n \} \\ & P\ 0 . \end{aligned}$$

Having done math. induction can be seen as having designed a *program that generates a proof*— given any $n :: Nat$ we can generate a proof of $P\ n$ in the manner above.

1.1.2 Inductively Definition of Functions

Inductively Defined Functions

- Since the type Nat is defined by two cases, it is natural to define functions on Nat following the structure:

$$\begin{aligned} exp & :: Nat \rightarrow Nat \rightarrow Nat \\ exp\ b\ 0 & = 1 \\ exp\ b\ (\mathbf{1}_+ n) & = b \times exp\ b\ n . \end{aligned}$$

- Even addition can be defined inductively

$$\begin{aligned} (+) & :: Nat \rightarrow Nat \rightarrow Nat \\ 0 + n & = n \\ (\mathbf{1}_+ m) + n & = \mathbf{1}_+ (m + n) . \end{aligned}$$

- Exercise: define (\times) ?

A Value Generator

Given the definition of exp , how does one compute $exp\ b\ 3$?

$$\begin{aligned} & exp\ b\ (\mathbf{1}_+ (\mathbf{1}_+ (\mathbf{1}_+ 0))) \\ = & \{ \text{definition of } exp \} \\ & b \times exp\ b\ (\mathbf{1}_+ (\mathbf{1}_+ 0)) \\ = & \{ \text{definition of } exp \} \\ & b \times b \times exp\ b\ (\mathbf{1}_+ 0) \\ = & \{ \text{definition of } exp \} \\ & b \times b \times b \times exp\ b\ 0 \\ = & \{ \text{definition of } exp \} \\ & b \times b \times b \times 1 . \end{aligned}$$

It is a program that generates a value, for any $n :: Nat$. Compare with the proof of P above.

Moral: Proving is Programming

An inductive proof is a program that generates a proof for any given natural number.

An inductive program follows the same structure of an inductive proof.

Proving and programming are very similar activities.

Without the $n + k$ Pattern

- Unfortunately, newer versions of Haskell abandoned the “ $n + k$ pattern” used in the previous slides:

$$\begin{aligned} exp & :: Int \rightarrow Int \rightarrow Int \\ exp\ b\ 0 & = 1 \\ exp\ b\ n & = b \times exp\ b\ (n - 1) . \end{aligned}$$

- Nat is defined to be Int in `MiniPrelude.hs`. Without `MiniPrelude.hs` you should use Int .

- For the purpose of this course, the pattern $\mathbf{1}_+ n$ reveals the correspondence between Nat and lists, and matches our proof style. Thus we will use it in the lecture.

- Remember to remove them in your code.

Proof by Induction

- To prove properties about Nat , we follow the structure as well.

- E.g. to prove that $exp\ b\ (m + n) = exp\ b\ m \times exp\ b\ n$.

- One possibility is to preform induction on m . That is, prove $P\ m$ for all $m :: Nat$, where $P\ m \equiv (\forall n :: exp\ b\ (m + n) = exp\ b\ m \times exp\ b\ n)$.

Case $m := 0$. For all n , we reason:

$$\begin{aligned} & exp\ b\ (0 + n) \\ = & \{ \text{defn. of } (+) \} \\ & exp\ b\ n \\ = & \{ \text{defn. of } (\times) \} \\ & 1 \times exp\ b\ n \\ = & \{ \text{defn. of } exp \} \\ & exp\ b\ 0 \times exp\ b\ n . \end{aligned}$$

We have thus proved $P\ 0$.

Case $m := \mathbf{1}_+ m$. For all n , we reason:

$$\begin{aligned}
 & \text{exp } b ((\mathbf{1}_+ m) + n) \\
 = & \quad \{ \text{defn. of } (+) \} \\
 & \text{exp } b (\mathbf{1}_+ (m + n)) \\
 = & \quad \{ \text{defn. of exp } \} \\
 & b \times \text{exp } b (m + n) \\
 = & \quad \{ \text{induction } \} \\
 & b \times (\text{exp } b m \times \text{exp } b n) \\
 = & \quad \{ (\times) \text{ associative } \} \\
 & (b \times \text{exp } b m) \times \text{exp } b n \\
 = & \quad \{ \text{defn. of exp } \} \\
 & \text{exp } b (\mathbf{1}_+ m) \times \text{exp } b n .
 \end{aligned}$$

We have thus proved $P (\mathbf{1}_+ m)$, given $P m$.

Structure Proofs by Programs

- The inductive proof could be carried out smoothly, because both $(+)$ and exp are defined inductively on its lefthand argument (of type Nat).
- The structure of the proof follows the structure of the program, which in turns follows the structure of the datatype the program is defined on.

Lists and Natural Numbers

- We have yet to prove that (\times) is associative.
- The proof is quite similar to the proof for associativity of $(++)$, which we will talk about later.
- In fact, Nat and lists are closely related in structure.
- Most of us are used to think of numbers as atomic and lists as structured data. Neither is necessarily true.
- For the rest of the course we will demonstrate induction using lists, while taking the properties for Nat as given.

1.1.3 A Set-Theoretic Explanation of Induction

An Inductively Defined Set?

- For a set to be “inductively defined”, we usually mean that it is the *smallest* fixed-point of some function.
- What does that mean?

Fixed-Point and Prefixed-Point

- A *fixed-point* of a function f is a value x such that $f x = x$.
- **Theorem.** f has fixed-point(s) if f is a *monotonic function* defined on a complete lattice.
 - In general, given f there may be more than one fixed-point.
- A *prefixed-point* of f is a value x such that $f x \leq x$.
 - Apparently, all fixed-points are also prefixed-points.
- **Theorem.** the smallest prefixed-point is also the smallest fixed-point.

Example: Nat

- Recall the usual definition: Nat is defined by the following rules:
 1. 0 is in Nat ;
 2. if n is in Nat , so is $\mathbf{1}_+ n$;
 3. there is no other Nat .
- If we define a function F from sets to sets: $F X = \{0\} \cup \{\mathbf{1}_+ n \mid n \in X\}$, 1. and 2. above means that $F \text{Nat} \subseteq \text{Nat}$. That is, Nat is a prefixed-point of F .
- 3. means that we want the *smallest* such prefixed-point.
- Thus Nat is also the least (smallest) fixed-point of F .

Least Prefixed-Point

Formally, let $F X = \{0\} \cup \{\mathbf{1}_+ n \mid n \in X\}$, Nat is a set such that

$$F \text{Nat} \subseteq \text{Nat} , \tag{1}$$

$$(\forall X : F X \subseteq X \Rightarrow \text{Nat} \subseteq X) , \tag{2}$$

where (1) says that Nat is a prefixed-point of F , and (2) it is the least among all prefixed-points of F .

Mathematical Induction, Formally

- Given property P , we also denote by P the set of elements that satisfy P .
- That $P 0$ and $P n \Rightarrow P (\mathbf{1}_+ n)$ is equivalent to $\{0\} \subseteq P$ and $\{\mathbf{1}_+ n \mid n \in P\} \subseteq P$,
- which is equivalent to $F P \subseteq P$. That is, P is a prefixed-point!

- By (2) we have $Nat \subseteq P$. That is, all Nat satisfy P !
- This is “why mathematical induction is correct.”

Coinduction?

There is a dual technique called *coinduction* where, instead of least prefixed-points, we talk about *greatest postfixed points*. That is, largest x such that $x \leq f x$.

With such construction we can talk about infinite data structures.

1.2 Induction on Lists

Inductively Defined Lists

- Recall that a (finite) list can be seen as a datatype defined by:²

$$\text{data List } a = [] \mid a : \text{List } a .$$

- Every list is built from the base case $[]$, with elements added by $(:)$ one by one: $[1, 2, 3] = 1 : (2 : (3 : []))$.

All Lists Today are Finite

But what about infinite lists?

- For now let’s consider finite lists only, as having infinite lists make the *semantics* much more complicated.³
- In fact, all functions we talk about today are total functions. No \perp involved.

Set-Theoretically Speaking...

The type $\text{List } a$ is the *smallest* set such that

1. $[]$ is in $\text{List } a$;
2. if xs is in $\text{List } a$ and x is in a , $x : xs$ is in $\text{List } a$ as well.

Inductively Defined Functions on Lists

- Many functions on lists can be defined according to how a list is defined:

$$\begin{aligned} \text{sum} &:: \text{List } Int \rightarrow Int \\ \text{sum } [] &= 0 \\ \text{sum } (x : xs) &= x + \text{sum } xs . \end{aligned}$$

$$\begin{aligned} \text{map} &:: (a \rightarrow b) \rightarrow \text{List } a \rightarrow \text{List } b \\ \text{map } f [] &= [] \\ \text{map } f (x : xs) &= f x : \text{map } f xs . \end{aligned}$$

$$\begin{aligned} - \text{sum } [1..10] &= 55 \\ - \text{map } (\mathbf{1}_+) [1, 2, 3, 4] &= [2, 3, 4, 5] \end{aligned}$$

²Not a real Haskell definition.

³What does that mean? Other courses in FLOLAC might cover semantics in more detail.

1.2.1 Append, and Some of Its Properties

List Append

- The function $(++)$ appends two lists into one

$$\begin{aligned} (++) &:: \text{List } a \rightarrow \text{List } a \rightarrow \text{List } a \\ [] ++ ys &= ys \\ (x : xs) ++ ys &= x : (xs ++ ys) . \end{aligned}$$

- Compare the definition with that of $(+)$!

Proof by Structural Induction on Lists

- Recall that every finite list is built from the base case $[]$, with elements added by $(:)$ one by one.
- To prove that some property P holds for all finite lists, we show that
 1. $P []$ holds;
 2. forall x and xs , $P (x : xs)$ holds provided that $P xs$ holds.

For a Particular List...

Given $P []$ and $P xs \Rightarrow P (x : xs)$, for all x and xs , how does one prove, for example, $P [1, 2, 3]$?

$$\begin{aligned} &P (1 : 2 : 3 : []) \\ \Leftarrow &\{ P (x : xs) \Leftarrow P xs \} \\ &P (2 : 3 : []) \\ \Leftarrow &\{ P (x : xs) \Leftarrow P xs \} \\ &P (3 : []) \\ \Leftarrow &\{ P (x : xs) \Leftarrow P xs \} \\ &P [] . \end{aligned}$$

Appending is Associative

To prove that $xs ++ (ys ++ zs) = (xs ++ ys) ++ zs$.

Let $P xs = (\forall ys, zs :: xs ++ (ys ++ zs) = (xs ++ ys) ++ zs)$, we prove P by induction on xs .

Case $xs := []$. For all ys and zs , we reason:

$$\begin{aligned} &[] ++ (ys ++ zs) \\ = &\{ \text{defn. of } (++) \} \\ &ys ++ zs \\ = &\{ \text{defn. of } (++) \} \\ &([] ++ ys) ++ zs . \end{aligned}$$

We have thus proved $P []$.

Case $xs := x : xs$. For all ys and zs , we reason:

$$\begin{aligned} &(x : xs) ++ (ys ++ zs) \\ = &\{ \text{defn. of } (++) \} \\ &x : (xs ++ (ys ++ zs)) \\ = &\{ \text{induction} \} \\ &x : ((xs ++ ys) ++ zs) \\ = &\{ \text{defn. of } (++) \} \\ &(x : (xs ++ ys)) ++ zs \\ = &\{ \text{defn. of } (++) \} \\ &((x : xs) ++ ys) ++ zs . \end{aligned}$$

We have thus proved $P(x : xs)$, given $P xs$.

Do We Have To Be So Formal?

- In our style of proof, every step is given a reason. Do we need to be so pedantic?
- Being formal *helps* you to do the proof:
 - In the proof of $\text{exp } b (m + n) = \text{exp } b m \times \text{exp } b n$, we expect that we will use induction to somewhere. Therefore the first part of the proof is to generate $\text{exp } b (m + n)$.
 - In the proof of associativity, we were working toward generating $xs ++ (ys ++ zs)$.
- By being formal we can work on the *form*, not the *meaning*. Like how we solved the knight/knave problem
- Being formal actually makes the proof easier!
- *Make the symbols do the work.*

Length

- The function *length* defined inductively:

$$\begin{aligned} \text{length} &:: \text{List } a \rightarrow \text{Nat} \\ \text{length } [] &= 0 \\ \text{length } (x : xs) &= \mathbf{1}_+ (\text{length } xs) . \end{aligned}$$

- Exercise: prove that *length* distributes into (*++*):

$$\text{length } (xs ++ ys) = \text{length } xs + \text{length } ys$$

Concatenation

- While (*++*) repeatedly applies (*:*), the function *concat* repeatedly calls (*++*):

$$\begin{aligned} \text{concat} &:: \text{List } (\text{List } a) \rightarrow \text{List } a \\ \text{concat } [] &= [] \\ \text{concat } (xs : xss) &= xs ++ \text{concat } xss . \end{aligned}$$

- Compare with *sum*.
- Exercise: prove $\text{sum} \cdot \text{concat} = \text{sum} \cdot \text{map } \text{sum}$.

1.2.2 More Inductively Defined Functions

Definition by Induction/Recursion

- Rather than giving commands, in functional programming we specify values; instead of performing repeated actions, we define values on inductively defined structures.

- Thus induction (or in general, recursion) is the only “control structure” we have. (We do identify and abstract over plenty of patterns of recursion, though.)

- **Note Terminology:** an inductive definition, as we have seen, define “bigger” things in terms of “smaller” things. Recursion, on the other hand, is a more general term, meaning “to define one entity in terms of itself.”
- To inductively define a function *f* on lists, we specify a value for the base case ($f []$) and, assuming that $f xs$ has been computed, consider how to construct $f (x : xs)$ out of $f xs$.

Filter

- *filter p xs* keeps only those elements in *xs* that satisfy *p*.

$$\begin{aligned} \text{filter} &:: (a \rightarrow \text{Bool}) \rightarrow \text{List } a \rightarrow \text{List } a \\ \text{filter } p [] &= [] \\ \text{filter } p (x : xs) &| p x = x : \text{filter } p xs \\ &| \text{otherwise} = \text{filter } p xs . \end{aligned}$$

Take and Drop

- Recall *take* and *drop*, which we used in the previous exercise.

$$\begin{aligned} \text{take} &:: \text{Nat} \rightarrow \text{List } a \rightarrow \text{List } a \\ \text{take } 0 \ xs &= [] \\ \text{take } (\mathbf{1}_+ n) [] &= [] \\ \text{take } (\mathbf{1}_+ n) (x : xs) &= x : \text{take } n \ xs . \end{aligned}$$

$$\begin{aligned} \text{drop} &:: \text{Nat} \rightarrow \text{List } a \rightarrow \text{List } a \\ \text{drop } 0 \ xs &= xs \\ \text{drop } (\mathbf{1}_+ n) [] &= [] \\ \text{drop } (\mathbf{1}_+ n) (x : xs) &= \text{drop } n \ xs . \end{aligned}$$

- Prove: $\text{take } n \ xs ++ \text{drop } n \ xs = xs$, for all *n* and *xs*.

TakeWhile and DropWhile

- *takeWhile p xs* yields the longest prefix of *xs* such that *p* holds for each element.

$$\begin{aligned} \text{takeWhile} &:: (a \rightarrow \text{Bool}) \rightarrow \text{List } a \rightarrow \text{List } a \\ \text{takeWhile } p [] &= [] \\ \text{takeWhile } p (x : xs) &| p x = x : \text{takeWhile } p \ xs \\ &| \text{otherwise} = [] . \end{aligned}$$

- *dropWhile p xs* drops the prefix from *xs*.

$$\begin{aligned} \text{dropWhile} &:: (a \rightarrow \text{Bool}) \rightarrow \text{List } a \rightarrow \text{List } a \\ \text{dropWhile } p [] &= [] \\ \text{dropWhile } p (x : xs) &| p x = \text{dropWhile } p \ xs \\ &| \text{otherwise} = x : xs . \end{aligned}$$

- Prove: $\text{takeWhile } p \ xs ++ \text{dropWhile } p \ xs = xs$.

List Reversal

- $reverse [1, 2, 3, 4] = [4, 3, 2, 1]$.

$$\begin{aligned} reverse & \quad \quad \quad :: List\ a \rightarrow List\ a \\ reverse [] & \quad \quad \quad = [] \\ reverse (x : xs) & = reverse\ xs ++ [x] . \end{aligned}$$

All Prefixes and Suffixes

- $inits [1, 2, 3] = [[], [1], [1, 2], [1, 2, 3]]$

$$\begin{aligned} inits & \quad \quad \quad :: List\ a \rightarrow List\ (List\ a) \\ inits [] & \quad \quad \quad = [[]] \\ inits (x : xs) & = [] : map\ (x :) (inits\ xs) . \end{aligned}$$

- $tails [1, 2, 3] = [[1, 2, 3], [2, 3], [3], []]$

$$\begin{aligned} tails & \quad \quad \quad :: List\ a \rightarrow List\ (List\ a) \\ tails [] & \quad \quad \quad = [[]] \\ tails (x : xs) & = (x : xs) : tails\ xs . \end{aligned}$$

Totality

- Structure of our definitions so far:

$$\begin{aligned} f [] & \quad \quad \quad = \dots \\ f (x : xs) & = \dots f\ xs \dots \end{aligned}$$

- Both the empty and the non-empty cases are covered, guaranteeing there is a matching clause for all inputs.
- The recursive call is made on a “smaller” argument, guaranteeing termination.
- Together they guarantee that every input is mapped to some output. Thus they define *total* functions on lists.

1.2.3 Other Patterns of Induction

Variations with the Base Case

- Some functions discriminate between several base cases. E.g.

$$\begin{aligned} fib & \quad \quad \quad :: Nat \rightarrow Nat \\ fib\ 0 & \quad \quad \quad = 0 \\ fib\ 1 & \quad \quad \quad = 1 \\ fib\ (2 + n) & = fib\ (1 + n) + fib\ n . \end{aligned}$$

- Some functions make more sense when it is defined only on non-empty lists:

$$\begin{aligned} f [x] & \quad \quad \quad = \dots \\ f (x : xs) & = \dots \end{aligned}$$

- What about totality?

- They are in fact functions defined on a different datatype:

$$\mathbf{data}\ List^+ a = Singleton\ a \mid a : List^+ a .$$

- We do not want to define *map*, *filter* again for $List^+ a$. Thus we reuse $List\ a$ and pretend that we were talking about $List^+ a$.
- It’s the same with *Nat*. We embedded *Nat* into *Int*.
- Ideally we’d like to have some form of *subtyping*. But that makes the type system more complex.

Lexicographic Induction

- It also occurs often that we perform *lexicographic induction* on multiple arguments: some arguments decrease in size, while others stay the same.
- E.g. the function *merge* merges two sorted lists into one sorted list:

$$\begin{aligned} merge & \quad \quad \quad :: List\ Int \rightarrow List\ Int \rightarrow List\ Int \\ merge [] [] & \quad \quad \quad = [] \\ merge [] (y : ys) & \quad \quad \quad = y : ys \\ merge (x : xs) [] & \quad \quad \quad = x : xs \\ merge (x : xs) (y : ys) & \mid x \leq y = x : merge\ xs\ (y : ys) \\ & \mid \mathbf{otherwise} = y : merge\ (x : xs)\ ys . \end{aligned}$$

Zip

Another example:

$$\begin{aligned} zip & :: List\ a \rightarrow List\ b \rightarrow List\ (a, b) \\ zip [] [] & \quad \quad \quad = [] \\ zip [] (y : ys) & \quad \quad \quad = [] \\ zip (x : xs) [] & \quad \quad \quad = [] \\ zip (x : xs) (y : ys) & = (x, y) : zip\ xs\ ys . \end{aligned}$$

Non-Structural Induction

- In most of the programs we’ve seen so far, the recursive call are made on direct sub-components of the input (e.g. $f (x : xs) = ..f\ xs..$). This is called *structural induction*.
 - It is relatively easy for compilers to recognise structural induction and determine that a program terminates.
- In fact, we can be sure that a program terminates if the arguments get “smaller” under some (well-founded) ordering.

Mergesort

- In the implementation of mergesort below, for example, the arguments always get smaller in size.

```
msort :: List Int → List Int
msort [] = []
msort [x] = [x]
msort xs = merge (msort ys) (msort zs) ,
  where n = length xs `div` 2
        ys = take n xs
        zs = drop n xs .
```

- What if we omit the case for $[x]$?
- If all cases are covered, and all recursive calls are applied to smaller arguments, the program defines a total function.

A Non-Terminating Definition

- Example of a function, where the argument to the recursive does not reduce in size:

```
f :: Int → Int
f 0 = 0
f n = f n .
```

- Certainly f is not a total function. Do such definitions “mean” something? We will talk about these later.

1.3 User Defined Inductive Datatypes

Internally Labelled Binary Trees

- This is a possible definition of internally labelled binary trees:

```
data ITree a = Null | Node a (ITree a) (ITree a) ,
```

- on which we may inductively define functions:

```
sumT :: ITree Nat → Nat
sumT Null = 0
sumT (Node x t u) = x + sumT t + sumT u .
```

Exercise: given $(\downarrow) :: Nat \rightarrow Nat \rightarrow Nat$, which yields the smaller one of its arguments, define the following functions

1. $minT :: Tree Nat \rightarrow Nat$, which computes the minimal element in a tree.
2. $mapT :: (a \rightarrow b) \rightarrow Tree a \rightarrow Tree b$, which applies the functional argument to each element in a tree.
3. Can you define (\downarrow) inductively on Nat ?⁴

⁴In the standard Haskell library, (\downarrow) is called min .

Induction Principle for *Tree*

- What is the induction principle for *Tree*?
- To prove that a predicate P on *Tree* holds for every tree, it is sufficient to show that
 1. P Null holds, and;
 2. for every x, t , and u , if $P t$ and $P u$ holds, $P (\text{Node } x t u)$ holds.
- Exercise: prove that for all n and t , $minT (\text{mapT } (n+) t) = n + minT t$. That is, $minT \cdot \text{mapT } (n+) = (n+) \cdot minT$.

Induction Principle for Other Types

- Recall that $\text{data Bool} = \text{False} \mid \text{True}$. Do we have an induction principle for *Bool*?
- To prove a predicate P on *Bool* holds for all booleans, it is sufficient to show that
 1. $P \text{ False}$ holds, and
 2. $P \text{ True}$ holds.
- Well, of course.
- What about $(A \times B)$? How to prove that a predicate P on $(A \times B)$ is always true?
- One may prove some property P_1 on A and some property P_2 on B , which together imply P .
- That does not say much. But the “induction principle” for products allows us to extract, from a proof of P , the proofs P_1 and P_2 .
- Every inductively defined datatype comes with its induction principle.
- We will come back to this point later.

2 Program Derivation

2.1 Some Comments on Efficiency

Data Representation

- So far we have (surprisingly) been talking about mathematics without much concern regarding efficiency. Time for a change.
- Take lists for example. Recall the definition: $\text{data List } a = [] \mid a : \text{List } a$.
- Our representation of lists is biased. The left most element can be fetched immediately.

- Thus, $(:)$, $head$, and $tail$ are constant-time operations, while $init$ and $last$ takes linear-time.
- In most implementations, the list is represented as a linked-list.

List Concatenation Takes Linear Time

- Recall $(++)$:

$$\begin{aligned} [] ++ ys &= ys \\ (x : xs) ++ ys &= x : (xs ++ ys) \end{aligned}$$

- Consider $[1, 2, 3] ++ [4, 5]$:

$$\begin{aligned} &(1 : 2 : 3 : []) ++ (4 : 5 : []) \\ &= 1 : ((2 : 3 : []) ++ (4 : 5 : [])) \\ &= 1 : 2 : ((3 : []) ++ (4 : 5 : [])) \\ &= 1 : 2 : 3 : ([] ++ (4 : 5 : [])) \\ &= 1 : 2 : 3 : 4 : 5 : [] \end{aligned}$$

- $(++)$ runs in time proportional to the length of its left argument.

Full Persistency

- Compound data structures, like simple values, are just values, and thus must be *fully persistent*.
- That is, in the following code:

```
let xs = [1, 2, 3]
    ys = [4, 5]
    zs = xs ++ ys
in ... body ...
```

- The *body* may have access to all three values. Thus $++$ cannot perform a destructive update.

Linked v.s. Block Data Structures

- Trees are usually represented in a similar manner, through links.
- Fully persistency is easier to achieve for such linked data structures.
- Accessing arbitrary elements, however, usually takes linear time.
- In imperative languages, constant-time random access is usually achieved by allocating lists (usually called arrays in this case) in a consecutive block of memory.

- Consider the following code, where xs is an array (implemented as a block), and ys is like xs , apart from its 10th element:

```
let xs = [1..100]
    ys = update xs 10 20
in ... body ...
```

- To allow access to both xs and ys in *body*, the *update* operation has to duplicate the entire array.
- Thus people have invented some smart data structure to do so, in around $O(\log n)$ time.
- On the other hand, *update* may simply overwrite xs if we can somehow make sure that *nobody* other than ys uses xs .
- Both are advanced topics, however.

Another Linear-Time Operation

- Taking all but the last element of a list:

$$\begin{aligned} init\ [x] &= [] \\ init\ (x : xs) &= x : init\ xs \end{aligned}$$

- Consider $init\ [1, 2, 3, 4]$:

$$\begin{aligned} &init\ (1 : 2 : 3 : 4 : []) \\ &= 1 : init\ (2 : 3 : 4 : []) \\ &= 1 : 2 : init\ (3 : 4 : []) \\ &= 1 : 2 : 3 : init\ (4 : []) \\ &= 1 : 2 : 3 : [] \end{aligned}$$

Sum, Map, etc

- Functions like *sum*, *maximum*, etc. needs to traverse through the list once to produce a result. So their running time is definitely $O(n)$, where n is the length of the list.
- If f takes time $O(t)$, *map f* takes time $O(n \times t)$ to complete. Similarly with *filter p*.
 - In a lazy setting, *map f* produces its first result in $O(t)$ time. We won't need lazy features for now, however.

2.2 Expand/Reduce Transformation

Sum of Squares

- Given a sequence a_1, a_2, \dots, a_n , compute $a_1^2 + a_2^2 + \dots + a_n^2$. Specification: *sumsq* = *sum* · *map square*.
- The spec. builds an intermediate list. Can we eliminate it?




Figure 1: How (++) allocates new (:) cells in the heap.

- The input is either empty or not. When it is empty:

$$\begin{aligned}
 & \text{sumsq []} \\
 = & \{ \text{definition of sumsq} \} \\
 & (\text{sum} \cdot \text{map square}) [] \\
 = & \{ \text{function composition} \} \\
 & \text{sum} (\text{map square} []) \\
 = & \{ \text{definition of map} \} \\
 & \text{sum} [] \\
 = & \{ \text{definition of sum} \} \\
 & 0
 \end{aligned}$$

Sum of Squares, the Inductive Case

- Consider the case when the input is not empty:

$$\begin{aligned}
 & \text{sumsq } (x : xs) \\
 = & \{ \text{definition of sumsq} \} \\
 & \text{sum} (\text{map square } (x : xs)) \\
 = & \{ \text{definition of map} \} \\
 & \text{sum} (\text{square } x : \text{map square } xs) \\
 = & \{ \text{definition of sum} \} \\
 & \text{square } x + \text{sum} (\text{map square } xs) \\
 = & \{ \text{definition of sumsq} \} \\
 & \text{square } x + \text{sumsq } xs
 \end{aligned}$$

Alternative Definition for sumsq

- From $\text{sumsq} = \text{sum} \cdot \text{map square}$, we have proved that

$$\begin{aligned}
 \text{sumsq} [] &= 0 \\
 \text{sumsq } (x : xs) &= \text{square } x + \text{sumsq } xs
 \end{aligned}$$

- Equivalently, we have shown that $\text{sum} \cdot \text{map square}$ is a solution of

$$\begin{aligned}
 f [] &= 0 \\
 f (x : xs) &= \text{square } x + f xs
 \end{aligned}$$

- However, the solution of the equations above is unique.
- Thus we can take it as another definition of sumsq . Denotationally it is the same function; operationally, it is (slightly) quicker.
- Exercise: try calculating an inductive definition of count .

Remark: Why Functional Programming?

- Time to muse on the merits of functional programming. Why functional programming?
 - Algebraic datatype? List comprehension? Lazy evaluation? Garbage collection? These are just language features that can be migrated.
 - No side effects.⁵ But why taking away a language feature?

- By being pure, we have a simpler semantics in which we are allowed to construct and reason about programs.
 - In an imperative language we do not even have $f\ 4 + f\ 4 = 2 \times f\ 4$.
- Ease of reasoning. That's the main benefit we get.

Example: Computing Polynomial

Given a list $as = [a_0, a_1, a_2 \dots a_n]$ and $x :: \text{Int}$, the aim is to compute:

$$a_0 + a_1x + a_2x^2 + \dots + a_nx^n.$$

This can be specified by

$$\text{poly } x\ as = \text{sum} (\text{zipWith } (\times)\ as (\text{iterate } (\times x)\ 1)) ,$$

where iterate can be defined by

$$\begin{aligned}
 \text{iterate} &:: (a \rightarrow a) \rightarrow a \rightarrow \text{List } a \\
 \text{iterate } f\ x &= x : \text{map } f (\text{iterate } f\ x) .
 \end{aligned}$$

Iterating a List

To get some intuition about iterate let us try expanding it:

⁵Unless introduced in disciplined ways. For example, through a monad.

$$\begin{aligned}
& \text{iterate } f \ x \\
= & \{ \text{definition of } \text{iterate} \} \\
& x : \text{map } f \ (\text{iterate } f \ x) \\
= & \{ \text{definition of } \text{map} \} \\
& x : \text{map } f \ (x : \text{map } f \ (\text{iterate } f \ x)) \\
= & \{ \text{map fusion} \} \\
& x : f \ x : \text{map } (f \cdot f) \ (\text{iterate } f \ x) \\
= & \{ \text{definitions of } \text{iterate} \text{ and } \text{map} \} \\
& x : f \ x : f \ (f \ x) : \text{map } (f \cdot f) \ (\text{map } f \ (\text{iterate } f \ x)) \\
= & \{ \text{map fusion} \} \\
& x : f \ x : f \ (f \ x) : \text{map } (f \cdot f \cdot f) \ (\text{iterate } f \ x) \dots
\end{aligned}$$

Zippping with a Binary Operator

While *iterate* generate a list, it is immediately truncated by *zipWith*:

$$\begin{aligned}
\text{zipWith} & :: (a \rightarrow b \rightarrow c) \rightarrow \\
& \text{List } a \rightarrow \text{List } b \rightarrow \text{List } c \\
\text{zipWith } (\oplus) \ [] & \quad \quad \quad = [] \\
\text{zipWith } (\oplus) \ (x : xs) \ [] & \quad \quad = [] \\
\text{zipWith } (\oplus) \ (x : xs) \ (y : ys) & = \\
& x \oplus y : \text{zipWith } (\oplus) \ xs \ ys \ .
\end{aligned}$$

Running the Specification

Try expanding *poly* $x \ [a, b, c, d]$, we get

$$\begin{aligned}
& \text{poly } x \ [a, b, c, d] \\
= & \text{sum } (\text{zipWith } (\times) \ [a, b, c, d] \ (\text{iterate } (\times x) \ 1)) \\
= & \{ \text{expanding } \text{iterate} \} \\
& \text{sum } (\text{zipWith } (\times) \ [a, b, c, d] \\
& \quad (1 : (1 \times x) : (1 \times x \times x) : (1 \times x \times x \times x) : \\
& \quad \quad \text{map } (\times x)^4 \ (\text{iterate } (\times x) \ 1))) \\
= & a + b \times x + c \times x \times x + d \times x \times x \times x \ .
\end{aligned}$$

where f^4 denotes $f \cdot f \cdot f \cdot f$.

As the list gets longer, we get more $(\times x)$ accumulating. Can we do better?

The main calculation

$$\begin{aligned}
& \text{poly } x \ (a : as) \\
= & \{ \text{definition of } \text{poly} \} \\
& \text{sum } (\text{zipWith } (\times) \ (a : as) \ (\text{iterate } (\times x) \ 1)) \\
= & \{ \text{definition of } \text{iterate} \} \\
& \text{sum } (\text{zipWith } (\times) \ (a : as) \\
& \quad (1 : \text{map } (\times x) \ (\text{iterate } (\times x) \ 1))) \\
= & \{ \text{definitions of } \text{zipWith} \text{ and } \text{sum} \} \\
& a + \text{sum } (\text{zipWith } (\times) \ as \\
& \quad (\text{map } (\times x) \ (\text{iterate } (\times x) \ 1))) \\
= & \{ \text{see below} \} \\
& a + \text{sum } (\text{map } (\times x) \ (\text{zipWith } (\times) \\
& \quad as \ (\text{iterate } (\times x) \ 1))) \\
= & \{ \text{sum} \cdot \text{map } (\times x) = (\times x) \cdot \text{sum} \} \\
& a + (\text{sum } (\text{zipWith } (\times) \ as \ (\text{iterate } (\times x) \ 1))) \times x \\
= & \{ \text{definition of } \text{poly} \} \\
& a + (\text{poly } x \ as) \times x \ .
\end{aligned}$$

Zip-Map Exchange

In the 4th step we used the property $\text{zipWith } (\times) \ as \cdot \text{map } (\times x) = \text{map } (\times x) \cdot \text{zipWith } (\times) \ as$.

It applies to any operator (\otimes) that is associative. For an intuitive understanding:

$$\begin{aligned}
& \text{zipWith } (\otimes) \ [a, b, c] \ (\text{map } (\otimes x) \ [d, e, f]) \\
= & [a \otimes (d \otimes x), b \otimes (e \otimes x), c \otimes (f \otimes x)] \\
= & \{ \text{associativity: } m \otimes (n \otimes k) = (m \otimes n) \otimes k \} \\
& [(a \otimes d) \otimes x, (b \otimes e) \otimes x, (c \otimes f) \otimes x] \\
= & \text{map } (\otimes x) \ (\text{zipWith } (\otimes) \ [a, b, c] \ [d, e, f]) \ .
\end{aligned}$$

We can do a formal proof if we want.

Distributivity

In the 5th step we used the property $\text{sum} \cdot \text{map } (\times x) = (\times x) \cdot \text{sum}$. For that we need distributivity between addition and multiplication.

We used that law to push *sum* to the right.

This is the crucial property that allows us to speed up *poly*: we are allowed to factor out common $(\times x)$.

Computing Polynomial

To conclude, we get:

$$\begin{aligned}
\text{poly } x \ [] & = 0 \\
\text{poly } x \ (a : as) & = a + (\text{poly } as) \times x \ ,
\end{aligned}$$

which uses a linear number of (\times) .

Let the Symbols Do the Work!

How do we know what laws to use or to assume?

By observing the form of the expressions. Let the symbols do the work.

2.3 Tupling

Steep Lists

- A *steep list* is a list in which every element is larger than the sum of those to its right:

$$\begin{aligned}
\text{steep} & \quad \quad \quad :: \text{List } \text{Int} \rightarrow \text{Bool} \\
\text{steep} \ [] & \quad \quad = \text{True} \\
\text{steep } (x : xs) & = \text{steep } xs \ \wedge \ x > \text{sum } xs.
\end{aligned}$$

- The definition above, if executed directly, is an $O(n^2)$ program. Can we do better?

- Just now we learned to construct a generalised function which takes more input. This time, we try the dual technique: to construct a function returning more results.

Generalise by Returning More

- Recall that $fst (a, b) = a$ and $snd (a, b) = b$.
- It is hard to quickly compute *steep* alone. But if we define

$$\begin{aligned} steepsum &:: List\ Int \rightarrow (Bool \times Int) \\ steepsum\ xs &= (steep\ xs, sum\ xs), \end{aligned}$$

- and manage to synthesise a quick definition of *steepsum*, we can implement *steep* by $steep = fst \cdot steepsum$.
- We again proceed by case analysis. Trivially,

$$steepsum\ [] = (True, 0).$$

Deriving for the Non-Empty Case

For the case for non-empty inputs:

$$\begin{aligned} &steepsum\ (x : xs) \\ = &\{ \text{definition of } steepsum \} \\ &(steep\ (x : xs), sum\ (x : xs)) \\ = &\{ \text{definitions of } steep \text{ and } sum \} \\ &(steep\ xs \wedge x > sum\ xs, x + sum\ xs) \\ = &\{ \text{extracting sub-expressions involving } xs \} \\ &\mathbf{let}\ (b, y) = (steep\ xs, sum\ xs) \\ &\mathbf{in}\ (b \wedge x > y, x + y) \\ = &\{ \text{definition of } steepsum \} \\ &\mathbf{let}\ (b, y) = steepsum\ xs \\ &\mathbf{in}\ (b \wedge x > y, x + y). \end{aligned}$$

Synthesised Program

We have thus come up with a $O(n)$ time program:

$$\begin{aligned} steep &= fst \cdot steepsum \\ steepsum\ [] &= (True, 0) \\ steepsum\ (x : xs) &= \mathbf{let}\ (b, y) = steepsum\ xs \\ &\quad \mathbf{in}\ (b \wedge x > y, x + y), \end{aligned}$$

Being Quicker by Doing More?

- A more generalised program can be implemented more efficiently?
 - A common phenomena! Sometimes the less general function cannot be implemented inductively at all!
 - It also often happens that a theorem needs to be generalised to be proved. We will see that later.
- An obvious question: how do we know what generalisation to pick?

- There is no easy answer — finding the right generalisation one of the most difficulty act in programming!
- Sometimes we simply generalise by examining the form of the formula.

2.4 Accumulating Parameters

Reversing a List

- The function *reverse* is defined by:

$$\begin{aligned} reverse\ [] &= [], \\ reverse\ (x : xs) &= reverse\ xs ++ [x]. \end{aligned}$$

- E.g. $reverse\ [1, 2, 3, 4] = ((([] ++ [4]) ++ [3]) ++ [2]) ++ [1] = [4, 3, 2, 1]$.
- But how about its time complexity? Since $(++)$ is $O(n)$, it takes $O(n^2)$ time to revert a list this way.
- Can we make it faster?

2.4.1 Fast List Reversal

Introducing an Accumulating Parameter

- Let us consider a generalisation of *reverse*. Define:

$$\begin{aligned} revcat &:: List\ a \rightarrow List\ a \rightarrow List\ a \\ revcat\ xs\ ys &= reverse\ xs ++ ys. \end{aligned}$$

- If we can construct a fast implementation of *revcat*, we can implement *reverse* by:

$$reverse\ xs = revcat\ xs\ [].$$

Reversing a List, Base Case

Let us use our old trick. Consider the case when *xs* is []:

$$\begin{aligned} &revcat\ []\ ys \\ = &\{ \text{definition of } revcat \} \\ &reverse\ [] ++ ys \\ = &\{ \text{definition of } reverse \} \\ &[] ++ ys \\ = &\{ \text{definition of } (++) \} \\ &ys. \end{aligned}$$

Reversing a List, Inductive Case

Case $x : xs$:

$$\begin{aligned}
& \text{revcat } (x : xs) \text{ } ys \\
= & \{ \text{definition of revcat} \} \\
& \text{reverse } (x : xs) ++ ys \\
= & \{ \text{definition of reverse} \} \\
& (\text{reverse } xs ++ [x]) ++ ys \\
= & \{ \text{since } (xs ++ ys) ++ zs = xs ++ (ys ++ zs) \} \\
& \text{reverse } xs ++ ([x] ++ ys) \\
= & \{ \text{definition of revcat} \} \\
& \text{revcat } xs (x : ys).
\end{aligned}$$

Linear-Time List Reversal

- We have therefore constructed an implementation of *revcat* which runs in linear time!

$$\begin{aligned}
\text{revcat } [] \text{ } ys &= ys \\
\text{revcat } (x : xs) \text{ } ys &= \text{revcat } xs (x : ys).
\end{aligned}$$

- A generalisation of *reverse* is easier to implement than *reverse* itself? How come?
- If you try to understand *revcat* operationally, it is not difficult to see how it works.
 - The partially reverted list is *accumulated* in *ys*.
 - The initial value of *ys* is set by *reverse xs = revcat xs []*.
 - Hmm... it is like a *loop*, isn't it?

2.4.2 Tail Recursion and Loops

Tracing Reverse

$$\begin{aligned}
& \text{reverse } [1, 2, 3, 4] \\
= & \text{revcat } [1, 2, 3, 4] [] \\
= & \text{revcat } [2, 3, 4] [1] \\
= & \text{revcat } [3, 4] [2, 1] \\
= & \text{revcat } [4] [3, 2, 1] \\
= & \text{revcat } [] [4, 3, 2, 1] \\
= & [4, 3, 2, 1]
\end{aligned}$$

$$\begin{aligned}
\text{reverse } xs &= \text{revcat } xs [] \\
\text{revcat } [] \text{ } ys &= ys \\
\text{revcat } (x : xs) \text{ } ys &= \text{revcat } xs (x : ys)
\end{aligned}$$

```

xs, ys ← XS, [];
while xs ≠ [] do
  xs, ys ← (tail xs), (head xs : ys);
return ys

```

Tail Recursion

- Tail recursion: a special case of recursion in which the last operation is the recursive call.

$$\begin{aligned}
f \ x_1 \ \dots \ x_n &= \{ \text{base case} \} \\
f \ x_1 \ \dots \ x_n &= f \ x'_1 \ \dots \ x'_n
\end{aligned}$$

- To implement general recursion, we need to keep a stack of return addresses. For tail recursion, we do not need such a stack.
- Tail recursive definitions are like loops. Each x_i is updated to x'_i in the next iteration of the loop.
- The first call to f sets up the initial values of each x_i .

Accumulating Parameters

- To efficiently perform a computation (e.g. *reverse xs*), we introduce a generalisation with an extra parameter, e.g.:

$$\text{revcat } xs \text{ } ys = \text{reverse } xs ++ ys.$$

- Try to derive an efficient implementation of the generalised function. The extra parameter is usually used to “accumulate” some results, hence the name.
 - To make the accumulation work, we usually need some kind of associativity.
- A technique useful for, but not limited to, constructing tail-recursive definition of functions.

Accumulating Parameter: Another Example

- Recall the “sum of squares” problem:

$$\begin{aligned}
\text{sumsq } [] &= 0 \\
\text{sumsq } (x : xs) &= \text{square } x + \text{sumsq } xs.
\end{aligned}$$

- The program still takes linear space (for the stack of return addresses). Let us construct a tail recursive auxiliary function.
- Introduce $\text{ssp } xs \ n = \text{sumsq } xs + n$.
- Initialisation: $\text{sumsq } xs = \text{ssp } xs \ 0$.
- Construct *ssp*:

$$\begin{aligned}
\text{ssp } [] \ n &= 0 + n = n \\
\text{ssp } (x : xs) \ n &= (\text{square } x + \text{sumsq } xs) + n \\
&= \text{sumsq } xs + (\text{square } x + n) \\
&= \text{ssp } xs (\text{square } x + n).
\end{aligned}$$

2.5 Conclusions

Conclusions

- Let the symbols do the work!
 - Algebraic manipulation helps us to separate the more mechanical parts of reasoning, from the parts that needs real innovation.
- For more examples of fun program calculation, see Bird [Bir10].
- For a more systematic study of algorithms using functional program reasoning, see Bird and Gibbons [BG20].

3 Folds On Lists

A Common Pattern We've Seen Many Times...

$$\begin{aligned} \text{sum } [] &= 0 \\ \text{sum } (x : xs) &= x + \text{sum } xs \end{aligned}$$

$$\begin{aligned} \text{length } [] &= 0 \\ \text{length } (x : xs) &= 1 + \text{length } xs \end{aligned}$$

$$\begin{aligned} \text{map } f [] &= [] \\ \text{map } f (x : xs) &= f x : \text{map } f xs \end{aligned}$$

This pattern is extracted and called *foldr*:

$$\begin{aligned} \text{foldr } f e [] &= e, \\ \text{foldr } f e (x : xs) &= f x (\text{foldr } f e xs). \end{aligned}$$

3.1 The Ubiquitous *foldr*

Replacing Constructors

$$\begin{aligned} \text{foldr } f e [] &= e \\ \text{foldr } f e (x : xs) &= f x (\text{foldr } f e xs) \end{aligned}$$

- One way to look at *foldr* $(\oplus) e$ is that it replaces $[]$ with e and $(:)$ with (\oplus) :

$$\begin{aligned} &\text{foldr } (\oplus) e [1, 2, 3, 4] \\ &= \text{foldr } (\oplus) e (1 : (2 : (3 : (4 : [])))) \\ &= 1 \oplus (2 \oplus (3 \oplus (4 \oplus e))). \end{aligned}$$

- $\text{sum} = \text{foldr } (+) 0$.
- $\text{length} = \text{foldr } (\lambda x n.1 + n) 0$.
- $\text{map } f = \text{foldr } (\lambda x xs.f x : xs) []$.
- One can see that $\text{id} = \text{foldr } (:) []$.

Some Trivial Folds on Lists

- Function *max* returns the least upper bound of elements in a list:

$$\begin{aligned} \text{max } [] &= -\infty, \\ \text{max } (x : xs) &= x \uparrow \text{max } xs. \end{aligned}$$

$$\text{max} = \text{foldr } (\uparrow) -\infty.$$

- This function is actually called *maximum* in the standard Haskell Prelude, while *max* returns the maximum between its two arguments. For brevity, we denote the former by *max* and the latter by (\uparrow) .

- Function *prod* returns the product of a list:

$$\begin{aligned} \text{prod } [] &= 1, \\ \text{prod } (x : xs) &= x \times \text{prod } xs. \end{aligned}$$

$$\text{prod} = \text{foldr } (\times) 1.$$

- Function *and* returns the conjunction of a list:

$$\begin{aligned} \text{and } [] &= \text{true}, \\ \text{and } (x : xs) &= x \wedge \text{and } xs. \end{aligned}$$

$$\text{and} = \text{foldr } (\wedge) \text{true}.$$

- Lets emphasise again that *id* on lists is a fold:

$$\begin{aligned} \text{id } [] &= [], \\ \text{id } (x : xs) &= x : \text{id } xs. \end{aligned}$$

$$\text{id} = \text{foldr } (:) [].$$

Some Functions We Have Seen...

- $(++ ys) = \text{foldr } (:) ys$.

$$\begin{aligned} (++) &:: [a] \rightarrow [a] \rightarrow [a] \\ [] ++ ys &= ys \\ (x : xs) ++ ys &= x : (xs ++ ys) . \end{aligned}$$

- $\text{concat} = \text{foldr } (++) []$.

$$\begin{aligned} \text{concat} &:: [[a]] \rightarrow [a] \\ \text{concat } [] &= [] \\ \text{concat } (xs : xss) &= xs ++ \text{concat } xss . \end{aligned}$$

Replacing Constructors

- Understanding *foldr* from its type. Recall

$$\mathbf{data} [a] = [] \mid a : [a] .$$

- Types of the two constructors: $[] :: [a]$, and $(:) :: a \rightarrow [a] \rightarrow [a]$.

- *foldr* replaces the constructors:

$$\begin{aligned} \text{foldr} &:: (a \rightarrow b \rightarrow b) \rightarrow b \rightarrow [a] \rightarrow b \\ \text{foldr } f e [] &= e \\ \text{foldr } f e (x : xs) &= f x (\text{foldr } f e xs) . \end{aligned}$$

3.2 The Fold-Fusion Theorem

Why Folds?

- “What are the three most important factors in a programming language?” Abstraction, abstraction, and abstraction!
- Control abstraction, procedure abstraction, data abstraction,...can programming patterns be abstracted too?
- Program structure becomes an entity we can talk about, reason about, and reuse.
 - We can describe algorithms in terms of fold, unfold, and other recognised patterns.
 - We can prove properties about folds,
 - and apply the proved theorems to all programs that are folds, either for compiler optimisation, or for mathematical reasoning.
- Among the theorems about folds, the most important is probably the *fold-fusion* theorem.

The Fold-Fusion Theorem

The theorem is about when the composition of a function and a fold can be expressed as a fold.

Theorem 1 (*foldr*-Fusion). Given $f :: a \rightarrow b \rightarrow b$, $e :: b$, $h :: b \rightarrow c$, and $g :: a \rightarrow c \rightarrow c$, we have:

$$h \cdot \text{foldr } f \ e = \text{foldr } g \ (h \ e) \ ,$$

if $h (f \ x \ y) = g \ x \ (h \ y)$ for all x and y .

For program derivation, we are usually given h , f , and e , from which we have to construct g .

Tracing an Example

Let us try to get an intuitive understand of the theorem:

$$\begin{aligned} & h (\text{foldr } f \ e \ [a, b, c]) \\ = & \{ \text{definition of } \text{foldr} \} \\ & h (f \ a \ (f \ b \ (f \ c \ e))) \\ = & \{ \text{since } h (f \ x \ y) = g \ x \ (h \ y) \} \\ & g \ a \ (h (f \ b \ (f \ c \ e))) \\ = & \{ \text{since } h (f \ x \ y) = g \ x \ (h \ y) \} \\ & g \ a \ (g \ b \ (h (f \ c \ e))) \\ = & \{ \text{since } h (f \ x \ y) = g \ x \ (h \ y) \} \\ & g \ a \ (g \ b \ (g \ c \ (h \ e))) \\ = & \{ \text{definition of } \text{foldr} \} \\ & \text{foldr } g \ (h \ e) \ [a, b, c] \ . \end{aligned}$$

Sum of Squares, Again

- Consider $\text{sum} \cdot \text{map square}$ again. This time we use the fact that $\text{map } f = \text{foldr } (mf \ f) \ []$, where $mf \ f \ x \ xs = f \ x : xs$.
- $\text{sum} \cdot \text{map square}$ is a fold, if we can find a ssq such that $\text{sum} (mf \ \text{square} \ x \ xs) = ssq \ x (\text{sum } xs)$. Let us try:

$$\begin{aligned} & \text{sum} (mf \ \text{square} \ x \ xs) \\ = & \{ \text{definition of } mf \} \\ & \text{sum} (\text{square} \ x : xs) \\ = & \{ \text{definition of } \text{sum} \} \\ & \text{square } x + \text{sum } xs \\ = & \{ \text{let } ssq \ x \ y = \text{square } x + y \} \\ & ssq \ x (\text{sum } xs) \ . \end{aligned}$$

Therefore, $\text{sum} \cdot \text{map square} = \text{foldr } ssq \ 0$.

Sum of Squares, without Folds

Recall that this is how we derived the inductive case of sumsq yesterday:

$$\begin{aligned} & \text{sumsq} (x : xs) \\ = & \{ \text{definition of } \text{sumsq} \} \\ & \text{sum} (\text{map square} (x : xs)) \\ = & \{ \text{definition of } \text{map} \} \\ & \text{sum} (\text{square } x : \text{map square } xs) \\ = & \{ \text{definition of } \text{sum} \} \\ & \text{square } x + \text{sum} (\text{map square } xs) \\ = & \{ \text{definition of } \text{sumsq} \} \\ & \text{square } x + \text{sumsq } xs \ . \end{aligned}$$

Comparing the two derivations, by using fold-fusion we supply only the “important” part.

More on Folds and Fold-fusion

- Compare the proof with the one yesterday. They are essentially the same proof.
- Fold-fusion theorem abstracts away the common parts in this kind of inductive proofs, so that we need to supply only the “important” parts.
- Tupling can be seen as a kind of fold-fusion. The derivation of steepsum , for example, can be seen as fusing:

$$\text{steepsum} \cdot \text{id} = \text{steepsum} \cdot \text{foldr } (:) \ [].$$

- Recall that $\text{steepsum } xs = (\text{steep } xs, \text{sum } xs)$. Reformulating steepsum into a fold allows us to compute it in one traversal.
- Not every function can be expressed as a fold. For example, $\text{tail} :: [a] \rightarrow [a]$ is not a fold!

3.3 More Useful Functions Defined as Folds

Longest Prefix

- The function call `takeWhile p xs` returns the longest prefix of `xs` that satisfies `p`:

```
takeWhile p [] = []
takeWhile p (x : xs) =
  if p x then x : takeWhile p xs
  else [] .
```

- E.g. `takeWhile (≤ 3) [1, 2, 3, 4, 5] = [1, 2, 3]`.
- It can be defined by a fold:

```
takeWhile p = foldr (take p) [],
take p x xs = if p x then x : xs else [] .
```

- Its dual, `dropWhile (≤ 3) [1, 2, 3, 4, 5] = [4, 5]`, is not a fold.

All Prefixes

- The function `inits` returns the list of all prefixes of the input list:

```
inits [] = [[]],
inits (x : xs) = [] : map (x :) (inits xs).
```

- E.g. `inits [1, 2, 3] = [[], [1], [1, 2], [1, 2, 3]]`.
- It can be defined by a fold:

```
inits = foldr ini [[]],
ini x xss = [] : map (x :) xss.
```

All Suffixes

- The function `tails` returns the list of all suffixes of the input list:

```
tails [] = [[]],
tails (x : xs) = let (ys : yss) = tails xs
  in (x : ys) : yss.
```

- E.g. `tails [1, 2, 3] = [[1, 2, 3], [2, 3], [3], []]`.
- It can be defined by a fold:

```
tails = foldr til [[]],
til x (ys : yss) = (x : ys) : yss.
```

- $scanr f e = map (foldr f e) \cdot tails$.

- E.g.

```
scanr (+) 0 [1, 2, 3]
= map sum (tails [1, 2, 3])
= map sum [[1, 2, 3], [2, 3], [3], []]
= [6, 5, 3, 0].
```

- Of course, it is slow to actually perform `map (foldr f e)` separately. By fold-fusion, we get a faster implementation:

```
scanr f e = foldr (sc f) [e],
sc f x (y : ys) = f x y : ys.
```

4 Folds on Other Algebraic Datatypes

- Folds are a specialised form of induction.
- Inductive datatypes: types on which you can perform induction.
- Every inductive datatype give rise to its fold.
- In fact, an inductive type can be defined by its fold.

Fold on Natural Numbers

- Recall the definition:

```
data Nat = 0 | 1+ Nat .
```

- Constructors: `0 :: Nat`, `(1+) :: Nat → Nat`.
- What is the fold on `Nat`?

```
foldN :: (a → a) → a → Nat → a
foldN f e 0 = e
foldN f e (1+ n) = f (foldN f e n) .
```

Examples of foldN

- $(+n) = foldN (1+) n$.
 $0 + n = n$
 $(1+ m) + n = 1+ (m + n)$.
- $(\times n) = foldN (n+) 0$.
 $0 \times n = 0$
 $(1+ m) \times n = n + (m \times n)$.
- $even = foldN not True$.
 $even 0 = True$
 $even (1+ n) = not (even n)$.

Fold-Fusion for Natural Numbers

Theorem 2 (*foldN*-Fusion). Given $f :: a \rightarrow a$, $e :: a$, $h :: a \rightarrow b$, and $g :: b \rightarrow b$, we have:

$$h \cdot \text{foldN } f \ e = \text{foldN } g \ (h \ e) ,$$

if $h (f \ x) = g (h \ x)$ for all x .

Exercise: fuse *even* into (+)?

Folds on Trees

- Example: internally labelled binary tree:

```
data ITree a = Null
             | Node a (ITree a) (ITree a) .
```

- Fold for ITree:

```
foldIT :: (a -> b -> b -> b) -> b -> ITree a -> b
foldIT f e Null           = e
foldIT f e (Node a t u) =
  f a (foldIT f e t) (foldIT f e u) .
```

Folds on Trees

- Example: externally labelled binary tree:
- Some datatypes for trees:

```
data ETree a = Tip a
             | Bin (ETree a) (ETree a) .
```

- Fold for ETree:

```
foldET :: (b -> b -> b) -> (a -> b)
        -> ETree a -> b
foldET f g (Tip x) = g x
foldET f g (Bin t u) =
  f (foldET f g t) (foldET f g u) .
```

Some Simple Functions on Trees

- To compute the size of an ITree:

$$\text{sizeIT} = \text{foldIT } (\lambda x \ m \ n \rightarrow \mathbf{1}_+ (m + n)) \ 0 .$$

- To sum up labels in an ETree:

$$\text{sizeET} = \text{foldET } (+) \ \text{id} .$$

- To compute a list of all labels in an ITree and an ETree:

```
flattenIT =
  foldIT (\x xs ys -> xs ++ [x] ++ ys) [] ,
flattenET = foldET (++) (\x -> [x]) .
```

- **Exercise:** what are the fusion theorems for *foldIT* and *foldET*?

5 Maximum Segment Sum

- The *maximum segment sum* is a classical problem, often used to demonstrate the effectiveness of program derivation.
- Given: a list of numbers — positive, zero, or negative.
- Compute: the maximum possible sum of a consecutive segment of the list.

Specifying Maximum Segment Sum

- A segment can be seen as a prefix of a suffix.
- The function *segs* computes the list of all the segments.

$$\text{segs} = \text{concat} \cdot \text{map } \text{inits} \cdot \text{tails} .$$

- Therefore, *mss* is specified by:

$$\text{mss} = \text{max} \cdot \text{map } \text{sum} \cdot \text{segs} .$$

The Derivation!

We reason:

$$\begin{aligned} & \text{max} \cdot \text{map } \text{sum} \cdot \text{concat} \cdot \text{map } \text{inits} \cdot \text{tails} \\ = & \{ \text{since } \text{map } f \cdot \text{concat} = \text{concat} \cdot \text{map } (\text{map } f) \} \\ & \text{map} \cdot \text{concat} \cdot \text{map } (\text{map } \text{sum}) \cdot \\ & \text{map } \text{inits} \cdot \text{tails} \\ = & \{ \text{since } \text{max} \cdot \text{concat} = \text{max} \cdot \text{map } \text{max} \} \\ & \text{max} \cdot \text{map } \text{max} \cdot \text{map } (\text{map } \text{sum}) \cdot \text{map } \text{inits} \cdot \text{tails} \\ = & \{ \text{since } \text{map } f \cdot \text{map } g = \text{map } (f \cdot g) \} \\ & \text{max} \cdot \text{map } (\text{max} \cdot \text{map } \text{sum} \cdot \text{inits}) \cdot \text{tails} . \end{aligned}$$

Recall the definition $\text{scanr } f \ e = \text{map } (\text{foldr } f \ e) \cdot \text{tails}$. If we can transform $\text{max} \cdot \text{map } \text{sum} \cdot \text{inits}$ into a fold, we can turn the algorithm into a *scanr*, which has a faster implementation.

Maximum Prefix Sum

Concentrate on $\text{max} \cdot \text{map } \text{sum} \cdot \text{inits}$:

$$\begin{aligned} & \text{max} \cdot \text{map } \text{sum} \cdot \text{inits} \\ = & \{ \text{def. of } \text{inits}, \text{ let } \text{ini } x \ \text{yss} = [] : \text{map } (x :) \ \text{yss} \} \\ & \text{max} \cdot \text{map } \text{sum} \cdot \text{foldr } \text{ini} \ [] [] \\ = & \{ \text{fold fusion, see below} \} \\ & \text{max} \cdot \text{foldr } \text{zplus} \ [0] . \end{aligned}$$

The fold fusion works because:

$$\begin{aligned} & \text{map } \text{sum} \ (\text{ini } x \ \text{yss}) \\ = & \text{map } \text{sum} \ ([] : \text{map } (x :) \ \text{yss}) \\ = & 0 : \text{map } (\text{sum} \cdot (x :)) \ \text{yss} \\ = & 0 : \text{map } (x+) \ (\text{map } \text{sum} \ \text{yss}) . \end{aligned}$$

Define $\text{zplus } x \ \text{yss} = 0 : \text{map } (x+) \ \text{yss}$.

Maximum Prefix Sum, 2nd Fold Fusion

Concentrate on $max \cdot map \ sum \cdot inits$:

$$\begin{aligned}
 & max \cdot map \ sum \cdot inits \\
 = & \{ \text{def. of } inits, \text{ let } ini \ x \ xss = [] : map \ (x:) \ xss \} \\
 & max \cdot map \ sum \cdot foldr \ ini \ [[]] \\
 = & \{ \text{fold fusion, } zplus \ x \ yss = 0 : map \ (x+) \ yss \} \\
 & max \cdot foldr \ zplus \ [0] \\
 = & \{ \text{fold fusion, let } zmax \ x \ y = 0 \ 'max' \ (x + y) \} \\
 & foldr \ zmax \ 0 \ .
 \end{aligned}$$

The fold fusion works because \uparrow distributes into $(+)$:

$$\begin{aligned}
 & max \ (0 : map \ (x+) \ xs) \\
 = & 0 \uparrow \ max \ (map \ (x+) \ xs) \\
 = & 0 \uparrow \ (x + max \ xs) \ .
 \end{aligned}$$

Back to Maximum Segment Sum

We reason:

$$\begin{aligned}
 & max \cdot map \ sum \cdot concat \cdot map \ inits \cdot tails \\
 = & \{ \text{since } map \ f \cdot concat = concat \cdot map \ (map \ f) \} \\
 & map \cdot concat \cdot map \ (map \ sum) \cdot \\
 & \quad map \ inits \cdot tails \\
 = & \{ \text{since } max \cdot concat = max \cdot map \ max \} \\
 & max \cdot map \ max \cdot map \ (map \ sum) \cdot \\
 & \quad map \ inits \cdot tails \\
 = & \{ \text{since } map \ f \cdot map \ g = map \ (f \cdot g) \} \\
 & max \cdot map \ (max \cdot map \ sum \cdot inits) \cdot tails \\
 = & \{ \text{previous reasoning} \} \\
 & max \cdot map \ (foldr \ zmax \ 0) \cdot tails \\
 = & \{ \text{introducing } scanr \} \\
 & max \cdot scanr \ zmax \ 0 \ .
 \end{aligned}$$

Maximum Segment Sum in Linear Time!

- We have derived $mss = max \cdot scanr \ zmax \ 0$, where $zmax \ x \ y = 0 \uparrow \ (x + y)$.
- The algorithm runs in linear time, but takes linear space.
- A tupling transformation eliminates the need for linear space.

$$mss = fst \cdot maxhd \cdot scanr \ zmax \ 0$$

where $maxhd \ xs = (max \ xs, head \ xs)$. We omit this last step in the lecture.

- The final program is $mss = fst \cdot foldr \ step \ (0, 0)$, where $step \ x \ (m, y) = ((0 \uparrow \ (x + y)) \uparrow \ m, 0 \uparrow \ (x + y))$.

6 Red-Black Tree

- A self-balancing binary search tree, often used to represent sets.
- Supports $O(\log n)$ -time searching, insertion, and deletion.
- One possible representation:

```

data RBTREE a = E |
              N Color (RBTREE a) a (RBTREE a) ,
data Color    = R | B .

```

Constraints

- It is a binary search tree.
 - In $N _ t \ x \ u$, every label in t is less than x , every label in u is more than x . The same holds for t and u .
- Each node is either colored red or black.
 - E is implicitly considered black.
- The root is black.
- Red nodes do not have red children.
- The number of black nodes from the root to each leaf is the same.

Searching

Searching in a red-black tree is just like that in a binary search tree:

```

search :: Int -> RBTREE Int -> Bool
search E           = False
search (N t x u) | k < x = ...
                  | k == x = ...
                  | k > x = ...

```

Exercise: what if we want to return the found element in a Maybe?

Insertion

- To insert a new element, perform a search to determine where to insert.
- The inserted node shall have color red.
- This would temporarily break the constraint that a red node shall not have a red children. We perform balancing upwards to restore the constraint. See the next slide.
- Finally we set the root to black.




Figure 2: Red-black tree balancing [Oka99].

Tree Balancing

- The re-balancing strategy is *not* unique.
- The strategy we will consider, shown in Figure 2, was presented by Okasaki [Oka99].
- Having only four rules, it is significantly simpler than those you'd find in most textbooks (which needs 8 rules or more)!
- Why?
- More will be discussed in the practicals.

[Hut16] Graham Hutton. *Programming in Haskell, 2nd Edition*. Cambridge University Press, 2016.

[Lip11] Miran Lipovača. *Learn You a Haskell for Great Good!* No Starch Press, 2011. Available online at <http://learnyouahaskell.com/>.

[Oka99] Chris Okasaki. Red-black trees in a functional setting. *Journal of Functional Programming*, 9(4):471–477, 1999.

[OSG98] Bryan O’Sullivan, Don Stewart, and John Goerzen. *Real World Haskell*. O’Reilly, 1998. Available online at <http://book.realworldhaskell.org/>.

References

[BG20] Richard S. Bird and Jeremy Gibbons. *Algorithm Design with Haskell*. Cambridge University Press, 2020.

[Bir98] Richard S. Bird. *Introduction to Functional Programming using Haskell*. Prentice Hall, 1998.

[Bir10] Richard S. Bird. *Pearls of Functional Algorithm Design*. Cambridge University Press, 2010.

A GHCi Commands

<code><statement></code>	evaluate/run <code><statement></code>
<code>:</code>	repeat last command
<code>:\{\n ..lines.. \n:\}\n}</code>	multiline command
<code>:add [*]<module> ...</code>	add module(s) to the current target set
<code>:browse[!] [[*]<mod>]</code>	display the names defined by module <code><mod></code> (!: more details; *: all top-level names)
<code>:cd <dir></code>	change directory to <code><dir></code>
<code>:cmd <expr></code>	run the commands returned by <code><expr>::IO String</code>
<code>:ctags[!] [<file>]</code>	create tags file for Vi (default: "tags") (!: use regex instead of line number)
<code>:def <cmd> <expr></code>	define command <code>:<cmd></code> (later defined command has precedence, <code>:<cmd></code> is always a builtin command)
<code>:edit <file></code>	edit file
<code>:edit</code>	edit last module
<code>:etags [<file>]</code>	create tags file for Emacs (default: "TAGS")
<code>:help, :?</code>	display this list of commands
<code>:info [<name> ...]</code>	display information about the given names
<code>:issafe [<mod>]</code>	display safe haskell information of module <code><mod></code>
<code>:kind <type></code>	show the kind of <code><type></code>
<code>:load [*]<module> ...</code>	load module(s) and their dependents
<code>:main [<arguments> ...]</code>	run the main function with the given arguments
<code>:module [+/-] [*]<mod> ...</code>	set the context for expression evaluation
<code>:quit</code>	exit GHCi
<code>:reload</code>	reload the current module set
<code>:run function [<arguments> ...]</code>	run the function with the given arguments
<code>:script <filename></code>	run the script <code><filename></code>
<code>:type <expr></code>	show the type of <code><expr></code>
<code>:undef <cmd></code>	undefine user-defined command <code>:<cmd></code>
<code>!<command></code>	run the shell command <code><command></code>

Commands for debugging

<code>:abandon</code>	at a breakpoint, abandon current computation
<code>:back</code>	go back in the history (after <code>:trace</code>)
<code>:break [<mod>] <l> [<col>]</code>	set a breakpoint at the specified location
<code>:break <name></code>	set a breakpoint on the specified function
<code>:continue</code>	resume after a breakpoint
<code>:delete <number></code>	delete the specified breakpoint
<code>:delete *</code>	delete all breakpoints
<code>:force <expr></code>	print <code><expr></code> , forcing unevaluated parts
<code>:forward</code>	go forward in the history (after <code>:back</code>)
<code>:history [<n>]</code>	after <code>:trace</code> , show the execution history
<code>:list</code>	show the source code around current breakpoint
<code>:list identifier</code>	show the source code for <code><identifier></code>
<code>:list [<module>] <line></code>	show the source code around line number <code><line></code>
<code>:print [<name> ...]</code>	prints a value without forcing its computation
<code>:sprint [<name> ...]</code>	simplified version of <code>:print</code>
<code>:step</code>	single-step after stopping at a breakpoint
<code>:step <expr></code>	single-step into <code><expr></code>
<code>:steplocal</code>	single-step within the current top-level binding
<code>:stepmodule</code>	single-step restricted to the current module
<code>:trace</code>	trace after stopping at a breakpoint

`:trace <expr>` evaluate `<expr>` with tracing on (see `:history`)

Commands for changing settings

`:set <option> ...` set options
`:seti <option> ...` set options for interactive evaluation only
`:set args <arg> ...` set the arguments returned by `System.getArgs`
`:set prog <progname>` set the value returned by `System.getProgName`
`:set prompt <prompt>` set the prompt used in GHCi
`:set editor <cmd>` set the command used for `:edit`
`:set stop [<n>] <cmd>` set the command to run when a breakpoint is hit
`:unset <option> ...` unset options

Options for `and`

`+m` allow multiline commands
`+r` revert top-level expressions after each evaluation
`+s` print timing/memory stats after each evaluation
`+t` print type after evaluation
`-<flags>` most GHC command line flags can also be set here (eg. `-v2`, `-fglasgow-exts`, etc). For GHCi-specific flags, see User's Guide, Flag reference, Interactive-mode options.

Commands for displaying information

`:show bindings` show the current bindings made at the prompt
`:show breaks` show the active breakpoints
`:show context` show the breakpoint context
`:show imports` show the current imports
`:show modules` show the currently loaded modules
`:show packages` show the currently active package flags
`:show language` show the currently active language flags
`:show <setting>` show value of `<setting>`, which is one of `[args, prog, prompt, editor, stop]`
`:showi language` show language flags for interactive evaluation